
File Names and Location

The latest versions of the Curtis-Godson transmittance tables are always located in the direc-
tory /hir1t/HIRDLS/HIRRAD/TRANSMITTANCE of the HIRDLS Science Computer Facility
(SCF).  There are two versions available, located in the subdirectories VERSION_0102 and
VERSION_0202.  The general form of the file names is

transtab_cCC_gGGGGGG_vVVVV.bin,

where CC is the two-digit channel number of the specific channel, GGGGGG is the six-character
species code of the specific gas, and VVVV is the four-digit version number of the specific table
version.  The possible values of the species code are: H2O---, CO2---, O3----, N2O---,
CH4---, O2----, SO2---, NO2---, HNO3--, CFC11-, CFC12-, CF4---, CLONO2, and
N2O5--.  The channel numbers are 01 to 21, and the version numbers are 0102 and 0202.
The last two digits of the version number indicate which version of the spectral response data
(see TC–NCA–082) was used to create the transmittance table; the first two digits indicate the
actual version of the table.  Version 1 was calculated from version 4 of the amount absorption
coefficient data files (see TC–NCA–081), and version 2 is a reformatted presentation of the
Curtis-Godson transmittance data that was calculated by Steven T. Massie at NCAR.

File Format

The transmittance tables are sequential, ANSI-IEEE binary files.  The specific structure of
the tables is illustrated by the following Fortran 95 input code sample:

implicit none

! Integer Parameters

! ------------------

! lnxi_unit = input unit number for the Curtis-Godson transmittance

!             table with the values of the natural logarithm of the

!             optical depth

integer(kind=4), parameter :: lnxi_unit = 1

! Integer Variables

! -----------------

! compression = data compression indicator, currently set to "0",

!               indicating no compression

!     N_sigma = number of wave number grid points

!         N_p = number of pressure grid points

!         N_T = number of temperature grid points

!       N_eta = number of column amount grid points

integer(kind=4) :: compression,N_sigma,N_p,N_T,N_eta

! Character Variables

! -------------------

! header_1 = first header of the Curtis-Godson transmittance table

! header_2 = second header of the Curtis-Godson transmittance table

! header_3 = third header of the Curtis-Godson transmittance table

! header_4 = fourth header of the Curtis-Godson transmittance table

character(len=80) :: header_1,header_2,header_3,header_4

! Real Variables

! --------------

! DELTAsigma = wave number grid size [cm-1]

!    lnp_max = maximum value of the natural logarithm of the pressure

!              grid (pressure "p" in hPa = mbar)

!   DELTAlnp = ln(p) grid size

!      T_min = minimum value of the temperature grid [K]

!     DELTAT = temperature grid size [K]

!  lneta_min = minimum value of the natural logarithm of the column

!              amount grid (column amount "eta" in mol / m2)

! DELTAlneta = ln(eta) grid size

!    sigma_l = wave number [cm-1] of the lower passband boundary

!       lnxi = natural logarithm of the gridded optical depth

real(kind=4) :: DELTAsigma,lnp_max,DELTAlnp,T_min,DELTAT,lneta_min,&

   DELTAlneta

real(kind=8) :: sigma_l

real(kind=4), allocatable :: lnxi(:,:,:)

open(unit=lnxi_unit,file='/hir1t/HIRDLS/HIRRAD/TRANSMITTANCE/&

   &VERSION_0102/transtab_c18_gH2O---_v0102.bin',status='old',&

   form='unformatted',action='read')

read(unit=lnxi_unit) header_1

read(unit=lnxi_unit) header_2

read(unit=lnxi_unit) header_3

read(unit=lnxi_unit) header_4

read(unit=lnxi_unit) compression,N_sigma,sigma_l,DELTAsigma,N_p,&

   lnp_max,DELTAlnp,N_T,T_min,DELTAT,N_eta,lneta_min,DELTAlneta

allocate (lnxi(N_p,N_T,N_eta))

read(unit=lnxi_unit) lnxi

close(unit=lnxi_unit,status='keep')

Data Parameters

For version 2 of the spectral response data, ∆σ  =  0.002 cm–1; and for version 1 of the
transmittance tables, the pressure and temperature grids are specified by:

Np  =  15,      (ln p)max  =  3 / lg e,      ∆(ln p)  =  –0.5 / lg e

and
NT  =  21,      Tmin  =  100 K,      ∆T  =  15 K;

where lg e is the common (decadic) logarithm of the natural base of logarithms.  The minimum
value and grid size of the column amount grid depend on the specific gas, but for all gases:

Nη  =  60.

The version 1 tables also have three dummy files for those gases that are not in a particular chan-
nel but are part of a joint retrieval in a set of channels, namely, CCl2F2 in channel 7, CCl3F in
channel 9, and N2O5 in channel 17.  For these dummy files:

Np  =  3,      (ln p)max  =  3 / lg e,      ∆(ln p)  =  –3 / lg e;

NT  =  3,      Tmin  =  100 K,      ∆T  =  150 K;

and
Nη  =  3,      (ln η)min  =  –100,      ∆(ln η)  =  75.

The optical depths in the dummy files are all set to a constant negligible value.
For version 2 of the transmittance tables:

Np  =  21,      (ln p)max  =  ln 500,      ∆(ln p)  =  (ln 2.54 × 10–5 – ln 500) / 20;

NT  =  22,      Tmin  =  150 K,      ∆T  =  10 K;

and
Nη  =  61.

The version 2 tables do not have dummy files, per se; the relevant gas-channel combinations
simply have the optical depths set to a constant negligible value.


